// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.2) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _ownerOf(tokenId);
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner or approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns the owner of the `tokenId`. Does NOT revert if token doesn't exist
     */
    function _ownerOf(uint256 tokenId) internal view virtual returns (address) {
        return _owners[tokenId];
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _ownerOf(tokenId) != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId, 1);

        // Check that tokenId was not minted by `_beforeTokenTransfer` hook
        require(!_exists(tokenId), "ERC721: token already minted");

        unchecked {
            // Will not overflow unless all 2**256 token ids are minted to the same owner.
            // Given that tokens are minted one by one, it is impossible in practice that
            // this ever happens. Might change if we allow batch minting.
            // The ERC fails to describe this case.
            _balances[to] += 1;
        }

        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId, 1);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     * This is an internal function that does not check if the sender is authorized to operate on the token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId, 1);

        // Update ownership in case tokenId was transferred by `_beforeTokenTransfer` hook
        owner = ERC721.ownerOf(tokenId);

        // Clear approvals
        delete _tokenApprovals[tokenId];

        unchecked {
            // Cannot overflow, as that would require more tokens to be burned/transferred
            // out than the owner initially received through minting and transferring in.
            _balances[owner] -= 1;
        }
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId, 1);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId, 1);

        // Check that tokenId was not transferred by `_beforeTokenTransfer` hook
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");

        // Clear approvals from the previous owner
        delete _tokenApprovals[tokenId];

        unchecked {
            // `_balances[from]` cannot overflow for the same reason as described in `_burn`:
            // `from`'s balance is the number of token held, which is at least one before the current
            // transfer.
            // `_balances[to]` could overflow in the conditions described in `_mint`. That would require
            // all 2**256 token ids to be minted, which in practice is impossible.
            _balances[from] -= 1;
            _balances[to] += 1;
        }
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId, 1);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens will be transferred to `to`.
     * - When `from` is zero, the tokens will be minted for `to`.
     * - When `to` is zero, ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 firstTokenId,
        uint256 batchSize
    ) internal virtual {}

    /**
     * @dev Hook that is called after any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens were transferred to `to`.
     * - When `from` is zero, the tokens were minted for `to`.
     * - When `to` is zero, ``from``'s tokens were burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 firstTokenId,
        uint256 batchSize
    ) internal virtual {}

    /**
     * @dev Unsafe write access to the balances, used by extensions that "mint" tokens using an {ownerOf} override.
     *
     * WARNING: Anyone calling this MUST ensure that the balances remain consistent with the ownership. The invariant
     * being that for any address `a` the value returned by `balanceOf(a)` must be equal to the number of tokens such
     * that `ownerOf(tokenId)` is `a`.
     */
    // solhint-disable-next-line func-name-mixedcase
    function __unsafe_increaseBalance(address account, uint256 amount) internal {
        _balances[account] += amount;
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/AdventStars.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity 0.8.22;

import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "./StarSky.sol";

/*
 *                               A celestial Advent Calendar,
 *                   to accompany its holder throughout the advent days.
 *
 *              Every day until Dec. 24th, a new star will shine in the night
 *                      sky, sometimes together with colorful nebulae.
 *
 *                 Each NFT will generate its own unique vibrant firmament.
 *                Only at Christmas Eve the cosmic artwork will be complete.
 *
 *                 Next year? Restart it, and the magic will happen again.
 *
 *                                                                           %@@-
 *                                                                          :%@@=
 *
 *
 *    @@
 *      @%
 *      @@
 *                                     .       . ...
 *                          . .          ... .              .@:.
 *                        . .   .
 *                    .     . . .      ..       .
 *      @                 .   . . . @ .    . .     .     .
 *               .    .  .  . . ...*@-:.:... ....     .    .. .
 *                  .-@- . .......     . .... ..  ..  . . .
 *               .           ............: .... .. ..
 *                  .. ..    .......... ..........   ..         ...    .               @@@
 *                 . . .. @@ ...:......    ..........    .              .
 *             .    .  .     ........::-%@:..........:.... ..    .
 *                ...............::.::.    .:..:.. .....  ..  ..
 *             .  ..  ........:.:.:::..::..:::..::.:..........  ...  ..
 *               . .  ....:..:..::.:.::..:.....::...:........ ...  ...   ..
 *                . .........:.:.::.:.::::...:...::::::.......... .   .    ..    .
 *                . ...........::::....::::---::::::.::::.:........:.... ....
 *            .   . ......:.:::.:::::::::.:...:.::::::::...::.:...... ...   ...  .      .
 *            .    ............::.:..:::::::::::::-.::::::.:.::.:::..:....  .  ..       .
 *             .    .............::::::::::    .::::::::.:.:.::::::.::. :-.... .  .
 *        .    .  ..  ......:::::..:.:::::: @@ -:::::::::-:::::::-::::::... ...:.  .
 *            . . ..  ..........:..:::.::::    ::::::::-:   :::-:::..:. :::.. ...
 *                  .  ....    .:.:..::::.::.::::::-::--- @ ---:---:.    ..:..:: .: .   .. .
 *           .    .  . .... @@ ..::::::::::-::::::-:--::-   --:-::---*@@=::::.::  :...
 *                .  . ....    ....:.::::::.:::::::--:----::::------:.-.::::-.:.:  ...   ..
 *            ..     . .  .........:.:...::::::--:--:----:   :-----:::...:.:-.::........ . .
 *                . .  .   ..........:.::::::::::-:----:--:@%------:-:::----:::::::.. ..      .
 *                    . ... ........:.:..:.:::::-:-:------:@=:-----=----:-:--:::::..-..:....
 *                      .  ..   ...........:::::::-:--:------:  :----:-----:-:-:::::..:.. :
 *                           .  .. . ..:.....::.:::-::    ----@@:--:-----:-:--::::::.::.: . ..
 *                        .   ..  ............::::-:-- @@ ---:=+.--:---:-:--.   -:::.:.:.. ..   ..
 *                          .       .  ..........::.::    ---::.:-------:---.=@ ::::::.:.....  .
 *              :            ..   .   .. ....:::::::::::-:::-:-::---:---:--::   ::.:-::: ..... .
 *                    .      .  .  ..    ...........::.:::-:------::::::   -:--::::::.:. ... . .
 *                                 . .... . ... ::-:::::.::::::::.::-----@:--:::.:::-..:...  . .
 *                                 .       . .::  ....:.::.:::-:-:-::..: ..::...::.-..      .
 *                            .     . .  .       .   :. .:.:....:.::--::-::..:----.: .:-+- ...
 *                                           .   .:..:-:...::::::::.::....::::..:.::..-#+::
 *                                             ..... .   ..  ....:.:.. .:.::: ::... ..
 *                                    @@   . .         .. . :.....  .::.:..:.... . .. ...  .
 *                                                 . .. .  ..  ...::::.        . .:...
 *                                                               .    .........
 *
 */

contract AdventStars is StarSky, ERC721, Ownable {
    uint256 public immutable MINT_END;
    uint256 public immutable PRICE;
    uint256 public constant MAX_SUPPLY = 432;

    uint256 currentToken = 1;

    mapping(uint256 => uint256) _seeds;
    mapping(uint256 => uint256) _tokenToYear;
    mapping(address => uint256) _discordUsersDiscount;

    constructor(
        uint256 price,
        uint256 mintEnd
    ) ERC721("Advent Stars", "STARS") {
        PRICE = price;
        MINT_END = mintEnd;
    }

    /* Admin */

    function withdraw() public onlyOwner {
        (bool success, ) = msg.sender.call{ value: address(this).balance }("");
        require(success, "fail");
    }

    function setDiscounts(
        address[] memory wallets,
        uint256[] memory discounts
    ) public onlyOwner {
        uint256 length = wallets.length;
        for (uint256 i = 0; i < length; ) {
            _discordUsersDiscount[wallets[i]] = discounts[i];
            unchecked {
                i++;
            }
        }
    }

    /* Public Write */

    function mint(uint256 amount) public payable {
        require(msg.value >= PRICE * amount, "not enough ether");
        require(block.timestamp < MINT_END, "mint ended");
        for (uint256 i = 0; i < amount; ) {
            _mint();
            unchecked {
                i++;
            }
        }
    }

    function restart(uint256 tokenId) public {
        require(msg.sender == ownerOf(tokenId), "not the owner");
        (uint256 month, uint256 day, uint256 year) = toDate(block.timestamp);
        require(2023 != year, "not this year");
        require(_tokenToYear[tokenId] != year, "already restarted");
        require((month == 11 && day > 23) || month == 12, "too early");

        _tokenToYear[tokenId] = year;
        _seeds[tokenId] = uint256(
            keccak256(abi.encodePacked(blockhash(block.number - 1), tokenId))
        );
    }

    function mintDiscount(uint256 amount) public payable {
        require(_discordUsersDiscount[msg.sender] > 0, "you have no discount");
        require(
            msg.value >=
                ((PRICE * amount) * 100) / _discordUsersDiscount[msg.sender],
            "not enough ether"
        );
        delete _discordUsersDiscount[msg.sender];
        require(block.timestamp < MINT_END, "mint ended");
        for (uint256 i = 0; i < amount; ) {
            _mint();
            unchecked {
                i++;
            }
        }
    }

    /* Public Read */

    function minted() public view returns (uint256) {
        return currentToken - 1;
    }

    function tokenAtIndex(uint256 index) public view returns (uint256) {
        for (uint256 i = 1; i < currentToken; ) {
            if (msg.sender == ownerOf(i)) {
                if (index == 0) {
                    return i;
                } else {
                    index--;
                }
            }

            unchecked {
                i = i + 1;
            }
        }

        revert("you don't that many tokens");
    }

    function adventDay() public view returns (uint256) {
        (uint256 month, uint256 day, ) = toDate(block.timestamp);

        if (month == 12 && day < 25) {
            return day;
        } else if (month == 12) {
            return 24;
        } else {
            return 0;
        }
    }

    function render(uint256 tokenId) public view returns (string memory) {
        (uint256 currentAdventDay, uint256 year) = _validateRequest(tokenId);
        return _render(_seeds[tokenId], currentAdventDay, year);
    }

    function tokenURI(
        uint256 tokenId
    ) public view override returns (string memory) {
        (uint256 currentAdventDay, uint256 year) = _validateRequest(tokenId);
        return _json(tokenId, _seeds[tokenId], currentAdventDay, year);
    }

    /* Possibly useful public utilities */

    function toDate(
        uint256 s
    ) public pure returns (uint256 month, uint256 day, uint256 year) {
        uint256 z = s / 86400 + 719468;
        uint256 era = (z >= 0 ? z : z - 146096) / 146097;
        uint256 doe = z - era * 146097;
        uint256 yoe = (doe - doe / 1460 + doe / 36524 - doe / 146096) / 365;
        year = yoe + era * 400;
        uint256 doy = doe - (365 * yoe + yoe / 4 - yoe / 100);
        uint256 mp = (5 * doy + 2) / 153;
        day = doy - (153 * mp + 2) / 5 + 1;
        month = uint256(int256(mp) + (mp < 10 ? int256(3) : -9));
        year += (month <= 2 ? 1 : 0);
    }

    /* Internal */

    function _validateRequest(
        uint256 tokenId
    ) internal view returns (uint256 currentAdventDay, uint256 tokenYear) {
        require(_exists(tokenId), "not a token");
        tokenYear = _tokenToYear[tokenId];
        if (tokenYear == 0) {
            tokenYear = 2023;
        }

        (uint256 month, uint256 day, uint256 currentYear) = toDate(
            block.timestamp
        );

        if (tokenYear == currentYear && month == 12 && day <= 24) {
            currentAdventDay = day;
        } else if (tokenYear == currentYear && month < 12) {
            currentAdventDay = 0;
        } else {
            currentAdventDay = 24;
        }
    }

    function _mint() internal {
        require(currentToken <= MAX_SUPPLY, "beyond supply");
        _seeds[currentToken] = uint256(
            keccak256(
                abi.encodePacked(blockhash(block.number - 1), currentToken)
            )
        );
        _mint(msg.sender, currentToken++);
    }
}

// File: contracts/DynamicBuffer.sol



// SPDX-License-Identifier: MIT
// Copyright (c) 2021 the ethier authors (github.com/divergencetech/ethier)

pragma solidity 0.8.22;

/// @title DynamicBuffer
/// @author David Huber (@cxkoda) and Simon Fremaux (@dievardump). See also
///         https://raw.githubusercontent.com/dievardump/solidity-dynamic-buffer
/// @notice This library is used to allocate a big amount of container memory
//          which will be subsequently filled without needing to reallocate
///         memory.
/// @dev First, allocate memory.
///      Then use `buffer.appendUnchecked(theBytes)` or `appendSafe()` if
///      bounds checking is required.
library DynamicBuffer {
    /// @notice Allocates container space for the DynamicBuffer
    /// @param capacity_ The intended max amount of bytes in the buffer
    /// @return buffer The memory location of the buffer
    /// @dev Allocates `capacity_ + 0x60` bytes of space
    ///      The buffer array starts at the first container data position,
    ///      (i.e. `buffer = container + 0x20`)
    function allocate(
        uint256 capacity_
    ) internal pure returns (bytes memory buffer) {
        assembly {
            // Get next-free memory address
            let container := mload(0x40)

            // Allocate memory by setting a new next-free address
            {
                // Add 2 x 32 bytes in size for the two length fields
                // Add 32 bytes safety space for 32B chunked copy
                let size := add(capacity_, 0x60)
                let newNextFree := add(container, size)
                mstore(0x40, newNextFree)
            }

            // Set the correct container length
            {
                let length := add(capacity_, 0x40)
                mstore(container, length)
            }

            // The buffer starts at idx 1 in the container (0 is length)
            buffer := add(container, 0x20)

            // Init content with length 0
            mstore(buffer, 0)
        }

        return buffer;
    }

    /// @notice Appends data to buffer, and update buffer length
    /// @param buffer the buffer to append the data to
    /// @param data the data to append
    /// @dev Does not perform out-of-bound checks (container capacity)
    ///      for efficiency.
    function appendUnchecked(
        bytes memory buffer,
        bytes memory data
    ) internal pure {
        assembly {
            let length := mload(data)
            for {
                data := add(data, 0x20)
                let dataEnd := add(data, length)
                let copyTo := add(buffer, add(mload(buffer), 0x20))
            } lt(data, dataEnd) {
                data := add(data, 0x20)
                copyTo := add(copyTo, 0x20)
            } {
                // Copy 32B chunks from data to buffer.
                // This may read over data array boundaries and copy invalid
                // bytes, which doesn't matter in the end since we will
                // later set the correct buffer length, and have allocated an
                // additional word to avoid buffer overflow.
                mstore(copyTo, mload(data))
            }

            // Update buffer length
            mstore(buffer, add(mload(buffer), length))
        }
    }

    /// @notice Appends data to buffer, and update buffer length
    /// @param buffer the buffer to append the data to
    /// @param data the data to append
    /// @dev Performs out-of-bound checks and calls `appendUnchecked`.
    function appendSafe(bytes memory buffer, bytes memory data) internal pure {
        checkOverflow(buffer, data.length);
        appendUnchecked(buffer, data);
    }

    /// @notice Appends data encoded as Base64 to buffer.
    /// @param fileSafe  Whether to replace '+' with '-' and '/' with '_'.
    /// @param noPadding Whether to strip away the padding.
    /// @dev Encodes `data` using the base64 encoding described in RFC 4648.
    /// See: https://datatracker.ietf.org/doc/html/rfc4648
    /// Author: Modified from Solady (https://github.com/vectorized/solady/blob/main/src/utils/Base64.sol)
    /// Author: Modified from Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/Base64.sol)
    /// Author: Modified from (https://github.com/Brechtpd/base64/blob/main/base64.sol) by Brecht Devos.
    function appendSafeBase64(
        bytes memory buffer,
        bytes memory data,
        bool fileSafe,
        bool noPadding
    ) internal pure {
        uint256 dataLength = data.length;

        if (data.length == 0) {
            return;
        }

        uint256 encodedLength;
        uint256 r;
        assembly {
            // For each 3 bytes block, we will have 4 bytes in the base64
            // encoding: `encodedLength = 4 * divCeil(dataLength, 3)`.
            // The `shl(2, ...)` is equivalent to multiplying by 4.
            encodedLength := shl(2, div(add(dataLength, 2), 3))

            r := mod(dataLength, 3)
            if noPadding {
                // if r == 0 => no modification
                // if r == 1 => encodedLength -= 2
                // if r == 2 => encodedLength -= 1
                encodedLength := sub(
                    encodedLength,
                    add(iszero(iszero(r)), eq(r, 1))
                )
            }
        }

        checkOverflow(buffer, encodedLength);

        assembly {
            let nextFree := mload(0x40)

            // Store the table into the scratch space.
            // Offsetted by -1 byte so that the `mload` will load the character.
            // We will rewrite the free memory pointer at `0x40` later with
            // the allocated size.
            mstore(0x1f, "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdef")
            mstore(
                0x3f,
                sub(
                    "ghijklmnopqrstuvwxyz0123456789-_",
                    // The magic constant 0x0230 will translate "-_" + "+/".
                    mul(iszero(fileSafe), 0x0230)
                )
            )

            // Skip the first slot, which stores the length.
            let ptr := add(add(buffer, 0x20), mload(buffer))
            let end := add(data, dataLength)

            // Run over the input, 3 bytes at a time.
            // prettier-ignore
            // solhint-disable-next-line no-empty-blocks
            for {} 1 {} {
                    data := add(data, 3) // Advance 3 bytes.
                    let input := mload(data)

                    // Write 4 bytes. Optimized for fewer stack operations.
                    mstore8(    ptr    , mload(and(shr(18, input), 0x3F)))
                    mstore8(add(ptr, 1), mload(and(shr(12, input), 0x3F)))
                    mstore8(add(ptr, 2), mload(and(shr( 6, input), 0x3F)))
                    mstore8(add(ptr, 3), mload(and(        input , 0x3F)))
                    
                    ptr := add(ptr, 4) // Advance 4 bytes.
                    // prettier-ignore
                    if iszero(lt(data, end)) { break }
                }

            if iszero(noPadding) {
                // Offset `ptr` and pad with '='. We can simply write over the end.
                mstore8(sub(ptr, iszero(iszero(r))), 0x3d) // Pad at `ptr - 1` if `r > 0`.
                mstore8(sub(ptr, shl(1, eq(r, 1))), 0x3d) // Pad at `ptr - 2` if `r == 1`.
            }

            mstore(buffer, add(mload(buffer), encodedLength))
            mstore(0x40, nextFree)
        }
    }

    /// @notice Returns the capacity of a given buffer.
    function capacity(bytes memory buffer) internal pure returns (uint256) {
        uint256 cap;
        assembly {
            cap := sub(mload(sub(buffer, 0x20)), 0x40)
        }
        return cap;
    }

    /// @notice Reverts if the buffer will overflow after appending a given
    /// number of bytes.
    function checkOverflow(
        bytes memory buffer,
        uint256 addedLength
    ) internal pure {
        uint256 cap = capacity(buffer);
        uint256 newLength = buffer.length + addedLength;
        if (cap < newLength) {
            revert("DynamicBuffer: Appending out of bounds.");
        }
    }
}

// File: contracts/StarSky.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.22;

import "@openzeppelin/contracts/utils/Strings.sol";
import "@openzeppelin/contracts/utils/Base64.sol";
import "./DynamicBuffer.sol";

contract StarSky {
    using Strings for uint8;
    using Strings for uint16;
    using Strings for uint256;

    string constant html1 =
        "<!DOCTYPE html><html lang='en'> <head> <meta charset='UTF-8'> <meta name='viewport' content='width=device-width, initial-scale=1.0, viewport-fit=cover'> <title>Advent Stars</title> <style> * { margin: 0; padding: 0; border: 0; } body { overflow: hidden; } </style></head><body>";
    string constant html2 = "</body></html>";

    bytes constant svg1 =
        "<svg xmlns='http://www.w3.org/2000/svg' viewBox='0 0 600 600' width='100%' height='100%'> <defs><style> @keyframes ifl { 0% { opacity: 1 } 40% { opacity: 0.3; }  100% { opacity: 0.8; } } #star { animation: ifl 20s infinite alternate-reverse; } #sky {transform-origin: center;} .year {font: 20px Times;fill: rgb(153,153,153);} .t {fill: transparent; } .sg { stroke: rgb(153,153,153) } .p { font: 25px Times; } .w { fill: white; }</style> <clipPath id='frame'><circle cx='300' cy='300' r='420' /></clipPath><filter id='blur' filterUnits='userSpaceOnUse' x='-50%' y='-50%' width='200%' height='200%'> <feGaussianBlur in='SourceGraphic' stdDeviation='5' result='blur5' /> <feGaussianBlur in='SourceGraphic' stdDeviation='10' result='blur10' /> <feGaussianBlur in='SourceGraphic' stdDeviation='20' result='blur30' /> <feMerge result='merged'> <feMergeNode in='blur10' /> <feMergeNode in='blur30' /> </feMerge> <feMerge > <feMergeNode in='blur5' /> <feMergeNode in='merged' /> </feMerge> </filter> ";
    bytes constant svg2 =
        " </defs> <rect x='0' y='0' width='600' height='600' class='sg' fill='black' />  <g id='sky' transform='rotate(";
    bytes constant svg3 = ") scale(0.7)' clip-path='url(#frame)'>";
    bytes constant svg4 =
        "</g> <circle cx='300' cy='300' r='294' class='t sg' />";
    bytes constant svg5 =
        "<line class='sg' x1='0' x2='40' y1='0' y2='40' /> <line class='sg' x1='600' x2='560' y1='0' y2='40' /> <line class='sg' x1='0' x2='40' y1='600' y2='560' /> <line class='sg' x1='600' x2='560' y1='600' y2='560' /></svg>";

    bytes constant text1 =
        "<circle cx='49' cy='51' r='13' class='t sg' /><text class='year' x='44' y='57'>";
    bytes constant text2 =
        "<circle cx='551' cy='51' r='13' class='t sg' /><text class='year' x='546' y='57'>";
    bytes constant text3 =
        "<circle cx='49' cy='549' r='13' class='t sg' /><text class='year' x='44' y='556'>";
    bytes constant text4 =
        "<circle cx='551' cy='549' r='13' class='t sg' /><text class='year' x='546' y='556'>";
    bytes constant textClose = "</text>";
    bytes constant frame =
        "<rect x='0' y='0' width='600' height='600' class='t sg' />";

    bytes constant placeholder0 =
        "<circle cx='300' cy='300' r='50' filter='url(#pb)' id='star' fill='hsl(";
    bytes constant placeholder1 =
        ",100%,60%)' /><circle cx='300' cy='300' r='30' filter='url(#pbc)' id='star' fill='rgba(255,255,255,0.5)' /><text text-anchor='middle' x='50%' y='47%' width='600' heigh='50' class='p w'>The first Star will appear on the</text> <text text-anchor='middle' x='50%' y='53%' width='600' heigh='50' class='p w'>first of December</text>";
    bytes constant placeholderBlur =
        "<filter id='pb' filterUnits='userSpaceOnUse' x='-50%' y='-50%' width='200%' height='200%'> <feGaussianBlur in='SourceGraphic' stdDeviation='50' result='b1' /> <feGaussianBlur in='SourceGraphic' stdDeviation='70' result='b2' /> <feGaussianBlur in='SourceGraphic' stdDeviation='120' result='b3' /> <feMerge result='m'> <feMergeNode in='b1' /> <feMergeNode in='b2' /> </feMerge> <feMerge> <feMergeNode in='b3' /> <feMergeNode in='m' /> </feMerge> </filter><filter id='pbc' filterUnits='userSpaceOnUse' x='-50%' y='-50%' width='200%' height='200%'> <feGaussianBlur in='SourceGraphic' stdDeviation='10' result='b1' /> <feGaussianBlur in='SourceGraphic' stdDeviation='30' result='b2' /> <feGaussianBlur in='SourceGraphic' stdDeviation='40' result='b3' /> <feMerge result='m'> <feMergeNode in='b1' /> <feMergeNode in='b2' /> </feMerge> <feMerge> <feMergeNode in='b3' /> <feMergeNode in='m' /> </feMerge> </filter>";

    bytes constant star0 = "<circle cx='";
    bytes constant star1 = "' cy='";
    bytes constant star2 = "' r='";
    bytes constant star3 = "' id='star' style='animation-duration:";
    bytes constant star4 = "s;' fill='rgba(";
    bytes constant starComma = ",";
    bytes constant star5 = ")' filter='url(#blur)' />";

    bytes constant starCore0 = "<circle cx='";
    bytes constant starCore1 = "' cy='";
    bytes constant starCore2 = "' r='";
    bytes constant starCore3 = "' class='w' />";

    bytes constant dustFilter0 = "<filter id='d";
    bytes constant dustFilter1 =
        "' filterUnits='userSpaceOnUse' x='-50%' y='-50%' width='200%' height='200%'> <feGaussianBlur in='SourceGraphic' stdDeviation='";
    bytes constant dustFilter2 =
        "' result='b1' /> <feGaussianBlur in='SourceGraphic' stdDeviation='";
    bytes constant dustFilter3 =
        "' result='b2' /> <feGaussianBlur in='SourceGraphic' stdDeviation='";
    bytes constant dustFilter4 =
        "' result='b3' /> <feMerge result='b'> <feMergeNode in='b1' /> <feMergeNode in='b2' /> <feMergeNode in='b3' /> </feMerge> <feColorMatrix result='cb' in='b' type='matrix' values=' ";
    bytes constant dustFilter5 = " 0 0 0 0 0 ";
    bytes constant dustFilter6 = " 0 0 0 0 0 ";
    bytes constant dustFilter7 = " 0 0 0 0 0 ";
    bytes constant dustFilter8 = " 0' /> </filter>";

    bytes constant dust0 = "<path d='M ";
    bytes constant dust1 = "' filter='url(#d";
    bytes constant dust2 = ")' stroke='white' stroke-width='";
    bytes constant dust3 = "px' />";

    uint256 constant STAR_TRAITS = 8;
    uint256 constant STAR_TRAIT_SIZE = 256 / STAR_TRAITS;
    uint256 constant STAR_TRAIT_MASK = 2 ** STAR_TRAIT_SIZE - 1;

    uint256 constant CONSTELLATION_TRAITS = 10;
    uint256 constant CONSTELLATION_TRAIT_SIZE = 256 / CONSTELLATION_TRAITS;
    uint256 constant CONSTELLATION_TRAIT_MASK =
        2 ** CONSTELLATION_TRAIT_SIZE - 1;

    struct Star {
        uint8 r;
        uint8 g;
        uint8 b;
        uint8 a;
        uint16 xRand;
        uint16 yRand;
        uint16 radius;
        uint16 duration;
        uint256 seed;
    }

    struct Constellation {
        bool incRand;
        uint8 keepProb;
        uint16 rotation;
        uint16 maxDust;
        uint16 startX;
        uint16 startY;
        uint16 minX;
        uint16 minY;
        uint16 maxX;
        uint16 maxY;
    }

    constructor() {}

    function _renderName(
        uint256 randomness
    ) internal pure returns (bytes memory) {
        uint256 lettersCount = (_starTrait(randomness, 0) % 4) + 1;
        uint256 numbersCount = (_starTrait(randomness, 1) % 5) + 1;
        bytes memory letters;
        for (uint8 i = 2; i < lettersCount + 2; i++) {
            letters = abi.encodePacked(
                letters,
                uint8((randomness >> i) % 25) + 65
            );
        }
        bytes memory numbers;
        for (uint8 i = 7; i < 7 + numbersCount; i++) {
            numbers = abi.encodePacked(
                numbers,
                uint8((randomness >> i) % 10) + 48
            );
        }

        return abi.encodePacked(letters, " ", numbers);
    }

    function _renderFloat(bytes memory buffer, uint16 number) internal pure {
        bytes memory numberStr = bytes(number.toString());
        if (numberStr.length == 4) {
            DynamicBuffer.appendUnchecked(
                buffer,
                abi.encodePacked(
                    numberStr[0],
                    numberStr[1],
                    ".",
                    numberStr[2],
                    numberStr[3]
                )
            );
        } else if (numberStr.length == 3) {
            DynamicBuffer.appendUnchecked(
                buffer,
                abi.encodePacked(numberStr[0], ".", numberStr[1], numberStr[2])
            );
        } else if (numberStr.length == 2) {
            DynamicBuffer.appendUnchecked(
                buffer,
                abi.encodePacked("0.", numberStr[0], numberStr[1])
            );
        } else {
            DynamicBuffer.appendUnchecked(
                buffer,
                abi.encodePacked("0.0", numberStr[0])
            );
        }
    }

    function _render(
        uint256 seed,
        uint256 day,
        uint256 year
    ) internal pure returns (string memory) {
        Constellation memory constellation = _constellation(seed);
        return string(_renderSVG(seed, day, year, constellation));
    }

    function _json(
        uint256 tokenId,
        uint256 seed,
        uint256 day,
        uint256 year
    ) internal pure returns (string memory) {
        Constellation memory constellation = _constellation(seed);
        bytes memory attributes = abi.encodePacked(
            '","attributes":',
            '[{"trait_type":"Cluster Density","value":"',
            (6 - constellation.keepProb).toString(),
            '"},{"trait_type":"Incremental","value":"',
            (constellation.incRand ? "True" : "False"),
            '"},{"trait_type":"Rotation","value":"',
            constellation.rotation.toString(),
            '"},{"trait_type":"Max Dust","value":"',
            constellation.maxDust.toString(),
            '"}]}'
        );

        bytes memory image = _renderSVG(seed, day, year, constellation);
        string memory imageAnimated = Base64.encode(image);
        image[188] = "c";
        string memory imageStatic = Base64.encode(image);

        bytes memory name = _renderName(seed);
        bytes memory description;
        if (day == 0) {
            description = abi.encodePacked(
                "The Star Cluster **",
                name,
                "** will start forming on 1st Dec. ",
                year.toString()
            );
        } else {
            description = abi.encodePacked(
                "View of the Star Cluster **",
                name,
                "** on ",
                day.toString(),
                "/12/",
                year.toString()
            );
        }
        return
            string(
                abi.encodePacked(
                    "data:application/json;base64,",
                    Base64.encode(
                        abi.encodePacked(
                            '{"name":"#',
                            tokenId.toString(),
                            " - ",
                            name,
                            '", "description":"',
                            description,
                            '","image":"data:image/svg+xml;base64,',
                            imageStatic,
                            '","animation_url":"data:image/svg+xml;base64,',
                            imageAnimated,
                            attributes
                        )
                    )
                )
            );
    }

    function _renderSVG(
        uint256 seed,
        uint256 day,
        uint256 year,
        Constellation memory constellation
    ) internal pure returns (bytes memory) {
        bytes memory starsRender = DynamicBuffer.allocate(100000);
        bytes memory filters = DynamicBuffer.allocate(100000);
        bytes memory dusts;
        bytes memory dustsFilters;

        if (day > 0) {
            uint16[2][25] memory points;
            Star[] memory stars = new Star[](25);
            uint8 i = 0;
            for (; i < day; i++) {
                uint256 seedRound = uint256(
                    keccak256(abi.encodePacked(seed, i + 1))
                );
                Star memory star = _decode(seedRound);
                stars[i] = star;

                uint16 newX;
                uint16 newY;

                if (constellation.incRand) {
                    newX = star.xRand;
                    newY = star.yRand;
                } else {
                    constellation.startX =
                        ((constellation.startX +
                            (star.xRand % constellation.maxX) +
                            constellation.minX) % 581) +
                        10;
                    constellation.startY =
                        ((constellation.startY +
                            (star.yRand % constellation.maxY) +
                            constellation.minY) % 581) +
                        10;
                    newX = constellation.startX;
                    newY = constellation.startY;
                }
                points[i] = [newX, newY];

                DynamicBuffer.appendUnchecked(starsRender, star0);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(newX.toString())
                );
                DynamicBuffer.appendUnchecked(starsRender, star1);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(newY.toString())
                );
                DynamicBuffer.appendUnchecked(starsRender, star2);
                _renderFloat(starsRender, star.radius + 300);
                DynamicBuffer.appendUnchecked(starsRender, star3);
                _renderFloat(starsRender, star.duration);
                DynamicBuffer.appendUnchecked(starsRender, star4);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(
                        (100 + (((uint256(star.r) * 1000) / 256) * 156) / 1000)
                            .toString()
                    )
                );
                DynamicBuffer.appendUnchecked(starsRender, starComma);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(
                        (100 + (((uint256(star.g) * 1000) / 256) * 156) / 1000)
                            .toString()
                    )
                );
                DynamicBuffer.appendUnchecked(starsRender, starComma);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(
                        (100 + (((uint256(star.b) * 1000) / 256) * 156) / 1000)
                            .toString()
                    )
                );
                DynamicBuffer.appendUnchecked(starsRender, starComma);
                _renderFloat(starsRender, star.a);
                DynamicBuffer.appendUnchecked(starsRender, star5);

                DynamicBuffer.appendUnchecked(starsRender, starCore0);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(newX.toString())
                );
                DynamicBuffer.appendUnchecked(starsRender, starCore1);
                DynamicBuffer.appendUnchecked(
                    starsRender,
                    bytes(newY.toString())
                );
                DynamicBuffer.appendUnchecked(starsRender, starCore2);
                _renderFloat(starsRender, star.radius);
                DynamicBuffer.appendUnchecked(starsRender, starCore3);
            }

            (dusts, dustsFilters) = _renderDust(
                points,
                i,
                stars,
                constellation
            );
        }

        bytes memory svg = DynamicBuffer.allocate(1000000);
        DynamicBuffer.appendUnchecked(svg, svg1);
        if (day == 0) {
            DynamicBuffer.appendUnchecked(svg, placeholderBlur);
        }
        DynamicBuffer.appendUnchecked(svg, filters);
        DynamicBuffer.appendUnchecked(svg, dustsFilters);
        DynamicBuffer.appendUnchecked(svg, svg2);
        DynamicBuffer.appendUnchecked(
            svg,
            bytes(constellation.rotation.toString())
        );
        DynamicBuffer.appendUnchecked(svg, svg3);
        DynamicBuffer.appendUnchecked(svg, dusts);
        DynamicBuffer.appendUnchecked(svg, starsRender);
        DynamicBuffer.appendUnchecked(svg, svg4);
        DynamicBuffer.appendUnchecked(svg, _renderYear(year, day));
        if (day == 0) {
            uint16 h = uint16(seed % 360);
            DynamicBuffer.appendUnchecked(svg, placeholder0);
            DynamicBuffer.appendUnchecked(svg, bytes(h.toString()));
            DynamicBuffer.appendUnchecked(svg, placeholder1);
        }
        DynamicBuffer.appendUnchecked(svg, svg5);

        return svg;
    }

    function _renderYear(
        uint256 year,
        uint256 day
    ) internal pure returns (bytes memory) {
        bytes memory yearBytes = bytes(year.toString());
        bytes memory dayBytes = bytes(day.toString());
        bytes memory text = DynamicBuffer.allocate(320);

        DynamicBuffer.appendUnchecked(text, text1);
        if (dayBytes.length == 2) {
            DynamicBuffer.appendUnchecked(text, abi.encodePacked(dayBytes[0]));
            DynamicBuffer.appendUnchecked(text, textClose);
            DynamicBuffer.appendUnchecked(text, text2);
            DynamicBuffer.appendUnchecked(text, abi.encodePacked(dayBytes[1]));
        } else {
            DynamicBuffer.appendUnchecked(text, bytes("0"));
            DynamicBuffer.appendUnchecked(text, textClose);
            DynamicBuffer.appendUnchecked(text, text2);
            DynamicBuffer.appendUnchecked(text, abi.encodePacked(dayBytes[0]));
        }
        DynamicBuffer.appendUnchecked(text, textClose);
        DynamicBuffer.appendUnchecked(text, text3);
        DynamicBuffer.appendUnchecked(text, abi.encodePacked(yearBytes[2]));
        DynamicBuffer.appendUnchecked(text, textClose);
        DynamicBuffer.appendUnchecked(text, text4);
        DynamicBuffer.appendUnchecked(text, abi.encodePacked(yearBytes[3]));
        DynamicBuffer.appendUnchecked(text, textClose);
        DynamicBuffer.appendUnchecked(text, frame);

        return text;
    }

    function _renderDust(
        uint16[2][25] memory points,
        uint8 length,
        Star[] memory stars,
        Constellation memory constellation
    ) internal pure returns (bytes memory, bytes memory) {
        bytes memory dusts = DynamicBuffer.allocate(100000);
        bytes memory dustsFilters = DynamicBuffer.allocate(100000);
        points = _sortPointsByDistance(points, length, [uint16(0), 0]);
        for (uint16 i = 0; i < length; i++) {
            uint16[2][25] memory subarray = createSubArray(points, length, i);
            uint16[2][25] memory sortedPoints = _sortPointsByDistance(
                subarray,
                length - i,
                points[i]
            );

            _buildDust(
                dusts,
                dustsFilters,
                sortedPoints,
                length - i,
                i,
                stars[i],
                constellation
            );
        }

        return (dusts, dustsFilters);
    }

    function _buildDust(
        bytes memory dusts,
        bytes memory dustsFilters,
        uint16[2][25] memory points,
        uint16 length,
        uint16 i,
        Star memory star,
        Constellation memory constellation
    ) internal pure {
        if (i % constellation.keepProb != 0) {
            return;
        }

        bytes memory pathPoints = DynamicBuffer.allocate(3200);
        _constructPath(pathPoints, points[0][0], points[0][1]);

        for (uint16 j = 1; j < length; j++) {
            uint16 diffX = absDiff(points[j][0], points[j - 1][0]);
            uint16 diffY = absDiff(points[j][1], points[j - 1][1]);

            if (diffX <= 200 && diffY <= 200) {
                _constructPath(pathPoints, points[j][0], points[j][1]);
            } else if (j > 3) {
                _constructDust(dusts, pathPoints, i, j, constellation);
                _constructDustsFilters(dustsFilters, i, star);
                break;
            } else {
                break;
            }
        }
    }

    function _constellationTrait(
        uint256 randomness,
        uint8 index
    ) internal pure returns (uint256) {
        return ((randomness >> (CONSTELLATION_TRAIT_SIZE * index)) &
            CONSTELLATION_TRAIT_MASK);
    }

    function _constellation(
        uint256 randomness
    ) internal pure returns (Constellation memory constellation) {
        constellation.rotation = uint16(
            _constellationTrait(randomness, 0) % 360
        );
        constellation.incRand = _constellationTrait(randomness, 1) % 2 == 0;
        constellation.maxDust = uint16(
            (_constellationTrait(randomness, 2) % 401) + 100
        );
        constellation.startX = uint16(
            (_constellationTrait(randomness, 3) % 581) + 10
        );
        constellation.startY = uint16(
            (_constellationTrait(randomness, 4) % 581) + 10
        );
        constellation.keepProb = uint8(
            (_constellationTrait(randomness, 5) % 6) + 1
        );
        constellation.minX = uint16(
            (_constellationTrait(randomness, 6) % 91) + 10
        );
        constellation.minY = uint16(
            (_constellationTrait(randomness, 7) % 91) + 10
        );
        constellation.maxX =
            uint16(
                (_constellationTrait(randomness, 8) %
                    (251 - constellation.minX))
            ) +
            1;
        constellation.maxY =
            uint16(
                (_constellationTrait(randomness, 9) %
                    (251 - constellation.minY))
            ) +
            1;
    }

    function absDiff(uint16 a, uint16 b) private pure returns (uint16) {
        if (a > b) {
            return a - b;
        } else {
            return b - a;
        }
    }

    function createSubArray(
        uint16[2][25] memory array,
        uint16 length,
        uint16 startIndex
    ) private pure returns (uint16[2][25] memory subArray) {
        uint16 newArrayLength = length - startIndex;
        for (uint16 i = 0; i < newArrayLength; i++) {
            subArray[i][0] = array[startIndex + i][0];
            subArray[i][1] = array[startIndex + i][1];
        }
    }

    function _constructPath(
        bytes memory buffer,
        uint16 x,
        uint16 y
    ) internal pure {
        DynamicBuffer.appendUnchecked(buffer, bytes(" "));
        DynamicBuffer.appendUnchecked(buffer, bytes(x.toString()));
        DynamicBuffer.appendUnchecked(buffer, bytes(","));
        DynamicBuffer.appendUnchecked(buffer, bytes(y.toString()));
    }

    function _constructDust(
        bytes memory buffer,
        bytes memory path,
        uint16 index,
        uint16 pathLength,
        Constellation memory constellation
    ) internal pure {
        uint16 thickness = constellation.maxDust / pathLength;

        DynamicBuffer.appendUnchecked(buffer, dust0);
        DynamicBuffer.appendUnchecked(buffer, path);
        DynamicBuffer.appendUnchecked(buffer, dust1);
        DynamicBuffer.appendUnchecked(buffer, bytes(index.toString()));
        DynamicBuffer.appendUnchecked(buffer, dust2);
        DynamicBuffer.appendUnchecked(buffer, bytes(thickness.toString()));
        DynamicBuffer.appendUnchecked(buffer, dust3);
    }

    function _constructDustsFilters(
        bytes memory buffer,
        uint16 index,
        Star memory star
    ) internal pure {
        uint16 baseDust = uint16((star.seed % 80) + 40);
        DynamicBuffer.appendUnchecked(buffer, dustFilter0);
        DynamicBuffer.appendUnchecked(buffer, bytes(index.toString()));
        DynamicBuffer.appendUnchecked(buffer, dustFilter1);
        DynamicBuffer.appendUnchecked(buffer, bytes(baseDust.toString()));
        DynamicBuffer.appendUnchecked(buffer, dustFilter2);
        DynamicBuffer.appendUnchecked(
            buffer,
            bytes((baseDust + 50).toString())
        );
        DynamicBuffer.appendUnchecked(buffer, dustFilter3);
        DynamicBuffer.appendUnchecked(
            buffer,
            bytes((baseDust + 100).toString())
        );
        DynamicBuffer.appendUnchecked(buffer, dustFilter4);
        _renderFloat(buffer, (uint16(star.r) * 100) / 256);
        DynamicBuffer.appendUnchecked(buffer, dustFilter5);
        _renderFloat(buffer, (uint16(star.g) * 100) / 256);
        DynamicBuffer.appendUnchecked(buffer, dustFilter6);
        _renderFloat(buffer, (uint16(star.b) * 100) / 256);
        DynamicBuffer.appendUnchecked(buffer, dustFilter7);
        _renderFloat(buffer, star.a);
        DynamicBuffer.appendUnchecked(buffer, dustFilter8);
    }

    function _sortPointsByDistance(
        uint16[2][25] memory points,
        uint16 length,
        uint16[2] memory origin
    ) internal pure returns (uint16[2][25] memory) {
        _quickSort(points, origin, 0, length - 1);
        return points;
    }

    function _quickSort(
        uint16[2][25] memory arr,
        uint16[2] memory origin,
        uint16 left,
        uint16 right
    ) internal pure {
        int16 i = int16(left);
        int16 j = int16(right);
        if (i == j) return;
        int256 pivot = _distanceSquared(
            arr[uint16(left + (right - left) / 2)],
            origin
        );
        while (i <= j) {
            while (_distanceSquared(arr[uint16(i)], origin) < pivot) i++;
            while (pivot < _distanceSquared(arr[uint16(j)], origin)) j--;
            if (i <= j) {
                (arr[uint16(i)], arr[uint16(j)]) = (
                    arr[uint16(j)],
                    arr[uint16(i)]
                );
                i++;
                j--;
            }
        }
        if (int16(left) < j) _quickSort(arr, origin, left, uint16(j));
        if (i < int16(right)) _quickSort(arr, origin, uint16(i), right);
    }

    function _distanceSquared(
        uint16[2] memory p,
        uint16[2] memory origin
    ) internal pure returns (int256) {
        return
            (int256(int16(p[0])) - int256(int16(origin[0]))) ** 2 +
            (int256(int16(p[1])) - int256(int16(origin[1]))) ** 2;
    }

    function _starTrait(
        uint256 randomness,
        uint8 index
    ) internal pure returns (uint256) {
        return ((randomness >> (STAR_TRAIT_SIZE * index)) & STAR_TRAIT_MASK);
    }

    function _decode(
        uint256 randomness
    ) internal pure returns (Star memory star) {
        star.r = uint8(_starTrait(randomness, 0) % 256);
        star.g = uint8(_starTrait(randomness, 1) % 256);
        star.b = uint8(_starTrait(randomness, 2) % 256);
        star.a = uint8((_starTrait(randomness, 3) % 101) + 1);
        star.xRand = uint16((_starTrait(randomness, 4) % 581) + 10);
        star.yRand = uint16((_starTrait(randomness, 5) % 581) + 10);
        star.radius = uint16((_starTrait(randomness, 6) % 500) + 100);
        star.duration = uint16((_starTrait(randomness, 7) % 1001) + 100);
        star.seed = randomness;
    }
}